*------------------------------------------------------------------------------*
* Family history - Dinas, Fouka, Schläpfer - JoP
*------------------------------------------------------------------------------*

*------------------------------------------------------------------------------*
* Figure 1
*------------------------------------------------------------------------------*



*------------------------------------------------------------------------------*
* Greece
*------------------------------------------------------------------------------*

use greece, clear

local cc=0
estimates clear
foreach x in rev1821st amdisasterst oxist civilwarst dictatorshipst {
	reg `x' asiaminor if T==0, cl(municipality)
	eststo m`cc'
	local cc=`cc'+1
}
	
coefplot (m0, offset(-0.01) aseq("Greek revolution") ciopts(lcolor(blue blue))) ///
(m1, aseq("Asia Minor") ciopts(lcolor(blue blue))) ///
(m2, aseq("WWII") ciopts(lcolor(blue blue))) ///
(m3, aseq("Civil war") ciopts(lcolor(blue blue))) ///
(m4, aseq("Dictatorship") ciopts(lcolor(blue blue))), ci(95 90) xline(0, lcolor(gs10)) keep(asiaminor)  ///
ysca(titlegap(2)) msize(small) mcolor(blue) ///
xtitle("", size(small)) ytitle("", size(small)) title("Greece", size(small)) ///
xlabel(, nogrid labsize(small)) ylabel(, nogrid labsize(small)) xsca(titlegap(2)) ///
swapnames legend(off) grid(none) xscale(range(-0.4(0.2)0.6))

graph save Greece, replace
 

*------------------------------------------------------------------------------*
* Germany
*------------------------------------------------------------------------------*


use germany, clear

local outcomes wwi_st reconstruction_st expellees_st wall_st
estimates clear

local cc=0
foreach var in `outcomes' {
	reg `var' expellee if control==1, cl(GKZ)
	eststo m`cc'
	local cc=`cc'+1
}

coefplot (m0, offset(-0.01) aseq("WWI") ciopts(lcolor(blue blue))) ///
(m2, aseq("Expellees") ciopts(lcolor(blue blue))) ///
(m1, aseq("Reconstruction") ciopts(lcolor(blue blue))) ///
(m3, aseq("Berlin wall") ciopts(lcolor(blue blue))), ci(95 90) xline(0, lcolor(gs10)) keep(expellee) ///
ysca(titlegap(2)) msize(small) mcolor(blue) ///
xtitle("", size(small)) ytitle("", size(small)) title("Germany", size(small)) ///
xlabel(, nogrid labsize(small)) ylabel(, nogrid labsize(small)) xsca(titlegap(2)) ///
swapnames legend(off) grid(none) xscale(range(-0.4(0.2)0.6))

graph save Germany, replace


*------------------------------------------------------------------------------*
* Combine
*------------------------------------------------------------------------------*


graph combine Greece.gph Germany.gph, xcommon iscale(1)





